
Equations and Codes for the PBPK Model 

1. Two-Compartment Dissolution Model 

The undissolved procaine penicillin G acts as the depot of penicillin G, and maintains the 

therapeutic concentration of penicillin G for at least 24 hours (Papich and Riviere, 2009; Uboh et 

al., 2000). The IM and SC injections were simulated using a two-compartment injection site 

model with a dissolution process based on the approach used to simulate intramuscular 

absorption of long-acting oxytetracycline (Lin et al., 2015). This approach divides penicillin G 

into dissolved penicillin G moieties and undissolved procaine penicillin G acting as depot 

(Figure 1). Equations S1-S7 that describe the process of IM injection are shown below: 

Rpen = Rinput * (1-Frac)        (S1) 

Rppg = Rinput * Frac        (S2) 

Rdiss = Appg*Kdiss         (S3) 

Rim = Kim*Apen         (S4) 

𝑑𝑑(𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴)/𝑑𝑑𝑑𝑑 = Rim         (S5) 

𝑑𝑑(𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴) 𝑑𝑑𝑑𝑑 = Rppg − Rdiss⁄        (S6) 

𝑑𝑑(𝐴𝐴𝐴𝐴𝐴𝐴𝐴𝐴) 𝑑𝑑𝑑𝑑⁄ = Rpen − Rim + Rdiss      (S7) 

where Rinput is the administration rate of IM injection (mg/h); Rpen is the administration rate of 

penicillin G moieties (mg/h); Rppg is the administration rate of procaine penicillin G (mg/h); 

Frac is the fraction of procaine penicillin G stayed undissolved (unitless); Kdiss is the dissolution 

rate constant of procaine penicillin G into penicillin G moieties (h−1); Rdiss is the dissolution rate 

of procaine penicillin G (mg/h); Rim is penicillin G absorption rate of IM route (mg/h); Appg is 

the amount of procaine penicillin G (mg); Apen is the amount of penicillin G moieties (mg); Kim 

is the penicillin G absorption rate constant of IM administration (h−1); Absorbim is the amount of 

penicillin G absorbed following IM injection (mg). 

2. Hepatic Metabolism 

A simplified first-order metabolic rate was used to simulate hepatic metabolism in this model 

(Fisher, 2000; Krishnan et al., 2009). The Equation S8 describing the process of first-order liver 

metabolism is shown below: 

Rmet = Kmet* CL*VL        (S8) 

where Rmet is the metabolic rate of penicillin G in liver (mg/h); Kmet is the metabolic rate 

constant of penicillin G (/h); CL is the penicillin G concentration in liver tissue (mg/L); VL is the 

tissue volume for liver (L). 

3. Urinary Elimination 

A first-order urinary elimination equation in the kidney compartment was adapted from previous 

research (Craigmill, 2003). The Equation S9 describing the process of first-order urinary 

elimination is shown below: 

Rurine = Kurine * CVK        (S9) 

where Rurine is the urine elimination rate of penicillin G (mg/h); Kurine is the urine elimination 

rate constant of penicillin G (L/h); CVK is the penicillin G concentration in kidney venous blood 

(mg/L). 

4. Mass Balance Equations in Kidney  

The rate of change for penicillin G in each tissue compartment was described using mass balance 

differential equations as we previously described (Leavens et al., 2012; Lin et al., 2015). Only 

penicillin G not bound to plasma proteins was considered as available for distribution. Kidney 

compartment was used as example for other tissue compartments. Equations S10-S11 for mass 

balance of penicillin G in flow-limited compartments using kidney as an example are described 

below. 

RK = QK * (CAfree - CVK) – Rurine      (S10) 

CVK = CK/PK         (S11) 

where RK is the rate of distribution of penicillin G in the kidney (mg/h); QK is the volume of 

blood flow to the kidney per hour (L/h); CAfree is the arterial blood concentration of penicillin G 

not bound with plasma proteins (mg/L); CVK is the concentration of penicillin G in the kidney 

venous blood (mg/L); Rurine is the urinary elimination rate of penicillin G (mg/h); CK is the 

penicillin G concentration in the kidney (mg/L); PK is the kidney:plasma partition coefficient 

(unitless). 

5. Sensitivity Analysis 

A local sensitivity analysis was performed to determine which parameters were most influential 

on the 24-hour AUC of penicillin G concentrations in plasma, liver, kidney, and muscle. Each 

parameter was increased by 1%, 5%, or 10% and the corresponding 24-hour AUC of penicillin 

concentrations were computed. Normalized sensitivity coefficient (NSC) was calculated using 

Equation S12 (Lin et al., 2011; Mirfazaelian et al., 2006) as shown below: 

NSC = Δr/r * p/Δp         (S12) 

where r is the response variable, and Δr is the change of the response variable resulting from 1%, 

5%, or 10% increase in the parameter value, p is the original value of the parameter of interest, 

Δp is 1%, 5%, or 10% of the original value of the parameter of interest. The relative influence of 

each parameter on the response variables was categorized as: low: |NSC|<0.2; medium: 

0.2≤|NSC|<0.5; high: 0.5≤|NSC| (Lin et al., 2013; Yoon et al., 2009).  

Supplementary Figures       
 

 

Figure S1. Calibration of the swine model. Comparison of model predictions (solid line) and observed data (blue squares) for penicillin G 

concentrations in the plasma, liver, fat and muscle of pigs exposed to procaine penicillin G via IM repeated 5 doses (65 mg/kg, A, B, C), IM 

repeated 3 doses (15 mg/kg, D, E), and single dose SC injection (11.7 mg/kg, F) are shown. Experimental data (mean ± SEM) are from previous 

studies (A-E) (Korsrud et al., 1998) and (F) (Ranheim et al., 2002). Limit of detection (LOD) is shown on each of the six panels using dotted line.  

LOD for the plasma is 1.5 ng/g, for the fat is 1.8 ng/g, for the liver is 1.8 ng/g, and for the muscle is 0.7 ng/g (Lupton et al., 2014). 

Figure S2. Calibration of the cattle model. Comparison of model predictions (solid line) and observed data (blue squares) for penicillin G 

concentrations in the plasma, muscle, and kidney of cattle exposed to procaine penicillin G via IM single dose (65 mg/kg, A), SC single dose (65 

mg/kg, B), IM with 5 repeated doses (24 mg/kg, C, E, F; 65 mg/kg, D, G, H) is shown. Experimental data for A, B, C, D (mean ± SEM) are from 

the previous study (Papich et al., 1993), and E, F, G, H (mean ± SEM) are from the study by Korsrud et al. (Korsrud et al., 1993). Tolerance 

(TOL) of penicillin G (0.05 µg/g) is shown on each of the panels using dotted line.   

Figure S3. Comparison of the changes of NSCs (normalized sensitivity coefficients) by 1%, 5% 

and 10% variations in parameter values. Only the sensitive parameters with at least one absolute 

value of NSC greater than 0.1 are plotted in the graph. Due to superimposing with other lines, the 

lines for the AUCCV are hard to be identified.

Supplementary Tables 
 

Table S1. Normalized sensitivity coefficients of representative parameters using AUCs for 
concentrations of penicillin G in plasma, liver, kidney and muscle as the dose metrics. 

 

Parameters AUCs of Penicillin Concentrations 
AUCCV AUCCL AUCCK AUCCM 

BW -0.10 -0.12 -0.10 -0.10 
QCC -0.39 -0.38 0.06 -0.39 
QKC -0.39 -0.39 0.06 -0.39 
VLC -0.11 -0.12 -0.11 -0.11 
Ksc 0.79 0.79 0.79 0.79 
Kim 0.41 0.41 0.41 0.41 
PL -0.11 0.88 -0.11 -0.11 
PK 0.00 0.00 1.00 0.00 
PM -0.01 -0.01 -0.01 0.99 

KmC -0.10 -0.12 -0.10 -0.10 
KurineC -0.46 -0.46 -0.92 -0.46 

 

Notes: Only parameters with at least one absolute value of NSC greater than 0.1 are shown in the table. 

AUCCV, AUCCL, AUCCK, and AUCCM represent 24-hour area under concentration curves of 

penicillin in plasma, liver, kidney and muscle. Please refer to Table 2 for abbreviations of parameters. 

PBPK Model Code 
Note: The Berkeley Madonna model code below is a general physiologically based pharmacokinetic 
(PBPK) model for procaine penicillin G in cattle and swine. Parameter values used in the model code are 
for procaine penicillin G in cattle. All parameter values in swine and cattle are summarized in Tables 2-3. 
Physiological parameter values of cattle and swine reported in our earlier paper by Lin et al. (2016) were 
for an average animal collected from previously published PBPK models. In the present penicillin G 
modeling work, in order to conduct population analysis, we need to have distributions of all parameters, 
so we conducted more extensive literature search on the physiological parameters of cattle and swine. As 
a result, some of the physiological parameters have been updated in this study. The updated physiological 
parameter values will be used in our subsequent models. Overall, the mean value of each physiological 
parameter is still quite close to the value reported in Lin et al. (2016) for both cattle and swine.  

{  
Penicillin PBPK model for Cattle (flow-limited model, linear metabolism equation, plasma protein binding) 
The PBPK model code is based on the Oxytetracycline.mmd from Lin et al. (2015) 
} 
 
METHOD RK4 
 
STARTTIME = 0 
STOPTIME= 100  ; h 
DT = 0.000391 
DTOUT = 0.1 
 
{Physiological Parameters} 
; Blood Flow Rates 
QCC = 5.97   ; L/h/kg, Cardiac Output (Doyle et al., 1960) 
 
; Fracion of blood flow to organs (unitless)  
QLC =  0.405   ; Fraction of blood flow to the liver (Doyle et al., 1960; Lescoat et al., 1996) 
QKC = 0.09   ; Fraction of blood flow to the kidneys (Lin et al., 2016) 
QFC = 0.08   ; Fraction of blood flow to the fat (Lin et al., 2016) 
QMC = 0.18   ; Fraction of blood flow to the muscle (Lin et al., 2016) 
QLuC = 1   ; Fraction of blood flow to the lung (Achenbach, 1995) 
QrestC = 1-QLC-QKC-QFC-QMC ; Fraction of blood flow to the rest of body (total sum equals to 1) 
 
; Tissue Volumes  
BW = 300   ; Body Weight (kg) (Mirzaei et al., 2011) 
; Fractional organ tissue volumes (unitless) 
VLC = 0.014   ; Fractional liver tissue (Swett et al., 1933) 
VKC = 0.002   ; Fractional kidney tissue (Swett et al., 1933) 
VFC = 0.15   ; Fractional fat tissue (Lin et al., 2016; Leavens et al., 2014) 
VMC = 0.27   ; Fractional muscle tissue (Lin et al., 2016; Leavens et al., 2014) 
VLuC = 0.008   ; Fractional lung tissue (Lin et al., 2016; Leavens et al., 2014; Swett et al., 1933) 
VvenC = 0.0296   ; Venous blood volume, fraction of blood volume (Lin et al., 2016) 
VartC = 0.0104   ; Arterial blood volume, fraction of blood volume (Lin et al., 2016) 
VrestC = 1-VLC-VKC-VFC-VMC-VLuC-VvenC-VartC; Fractional rest of body (total sum equals to 1) 
 
{Mass Transfer Parameters (Chemical-Specific Parameters)} 
; Partition Coefficients (PC, tissue:plasma) 
PL = 3    ; Liver:plasma PC (0.157, Tsuji et al., 1983, Table 4, in rats) 
PM = 0.3   ; Muscle:plasma PC (0.062, Tsuji et al., 1983, Table 4, in rats) 
PF = 0.04   ; Fat:plasma PC (0.062, adapted from muscle partition coefficient)  
PK = 2.5    ; Kidney:plasma PC (3.70, Tsuji et al., 1983, Table 4, in rats) 

PLu = 0.18   ; Lung:plasma PC (0.157, Tsuji et al., 1983, Table 4, in rats) 
Prest = 0.479   ; Rest of body:plasma PC (Cao et al. 2012, Table 1, estimated value in human) 
 
{Kinetic Constants} 
; IM Absorption Rate Constants 
Kim = 0.07   ; /h, IM absorption rate constant  
 
; SC Absorption Rate Constants 
Ksc = 0.02    ; /h, SC absorption rate constant 
 
; Percentage Plasma Protein Binding unitless 
PB = 0.483   ; Percentage of drug bound to plasma proteins (Keen, 1965) 
Free = 1-PB 
 
; Urinary Elimination Rate Constants 
KurineC = 0.45    ; L/h/kg  
 
; Metabolic Rate Constant 
KmC = 0.0025   ; /h/kg 
 
{Parameters for Various Exposure Scenarios} 
PDOSEsc = 0   ; (mg/kg) 
PDOSEim = 65   ; (mg/kg) 
 
{Cardiac output and blood flow to tissues (L/h)} 
QC = QCC*BW    ; Cardiac output 
QL = QLC*QC    ; Liver 
QK = QKC*QC    ; Kidney 
QF = QFC*QC    ; Fat 
QM = QMC*QC    ; Muscle 
QLu = QLuC*QC   ; Lung 
QR = QrestC*QC   ; Rest of body 
 
{Tissue volumes (L)} 
VL = VLC*BW    ; Liver 
VK = VKC*BW    ; Kidney 
VF = VFC*BW    ; Fat 
VM = VMC*BW   ; Muscle 
VLu = VLuC*BW   ; Lung 
VR = VrestC*BW   ; Rest of body 
Vven = VvenC*BW   ; Venous Blood 
Vart = VartC*BW   ; Arterial Blood 
 
; Urinary Elimination Rate Constants 
Kurine = KurineC*BW 
 
; Metabolic Rate Constant 
Kmet = KmC*BW    
 
{Dosing} 
; Dosing caculation based on BW 
DOSEsc = PDOSEsc*BW  ; (mg) 
DOSEim = PDOSEim*BW  ; (mg) 
 
Kdiss = 1e-5    ; /h 
Kdisssc = 1e-4   ; /h 

Frac = 0.6 
 
; Dosing, repeated doses 
tinterval = 24   ; Varied dependent on the exposure paradigm (h) 
Tdoses = 5    ; times for multiple oral gavage 
 
dosingperiod = if time < Tdoses*tinterval-DT then 1 else 0 
 
; Dosing, IM, intramuscular 
Rinputim = pulse(DOSEim,0,tinterval)*dosingperiod 
Rpenim = Rinputim*(1-Frac); 
Rppgim =  Rinputim*Frac; 
Rim = Kim*Amtsiteim 
d/dt(Absorbim) = Rim 
init Absorbim = 0 
d/dt(Amtsiteim) = Rpenim- Rim + Kdiss* DOSEppgim 
init Amtsiteim = 0 
d/dt(DOSEppgim) = Rppgim-Kdiss* DOSEppgim 
init DOSEppgim = 0 
 
; Dosing, SC, subcutaneous 
Rinputsc = pulse(DOSEsc,0,tinterval)*dosingperiod 
Rpensc = Rinputsc*(1-Frac); 
Rppgsc =  Rinputsc*Frac; 
Rsc = Ksc*Amtsitesc 
d/dt(Absorbsc) = Rsc 
init Absorbsc = 0 
d/dt(Amtsitesc) = Rpensc- Rsc + Kdisssc* DOSEppgsc 
init Amtsitesc = 0 
d/dt(DOSEppgsc) = Rppgsc-Kdisssc* DOSEppgsc 
init DOSEppgsc = 0 
 
{Penicillin distribution in each compartment} 
; Penicillin in venous blood compartment 
RV = (QL*CVL+QK*CVK+QF*CVF+QM*CVM+QR*CVR+Rsc+Rim)-QC*CV; RV the changing rate in the 
venous blood (mg/h) 
d/dt(AV) = RV   ; AV the amount of the drug in the venous blood (mg) 
init AV = 0 
CV = AV/Vven   ; CV drug concentration in the venous blood (mg/L) 
RA = QC*(CVLu-CAfree)  ; RA the changing rate in the arterial blood (mg/h) 
d/dt(AA) = RA 
init AA = 0   ; AA the amount of the drug in the arterial blood (mg) 
CA = AA/Vart   ; CAfree concentration of unbound drug in the arterial blood (mg/L) 
CAfree = CA*Free 
d/dt(AUCCV) = CV  ; AUCCV AUC of drug concentration in the venous blood (mg*h/L) 
init AUCCV = 0 
 
ABlood = AA+AV 
 
; Penicillin in liver compartment, flow-limited model 
RL = QL*(CAfree-CVL)-Rmet  ; RL the changing rate of the amount of drug in liver (mg/h) 
d/dt(AL) = RL    ; AL amount of drug in liver (mg)  
init AL = 0   
CL = AL/VL   ; CL drug concentration in liver (mg/L) 
CVL= AL/(VL*PL)  ; CVL drug concentration in venous blood from liver (mg/L) 
d/dt(AUCCL) = CL  ; AUCCL area under the curve of drug concentration in liver (mg*h/L) 

init AUCCL = 0 
 
; Metabolism of Penicillin in liver compartment 
Rmet = Kmet*CL*VL   ; Rmet the metabolic rate in liver (mg/h) 
d/dt(Amet) = Rmet  ; Amet the amount of drug metabolized in liver (mg) 
init Amet = 0 
 
; Penicillin in kidney compartment, flow-limited model 
RK = QK*(CAfree-CVK)-Rurine ; RK the changing rate of the amount of drug in kidney (mg/h) 
d/dt(AK) = RK   ; AK amount of drug in kidney (mg) 
init AK = 0   
CK = AK/VK   ; CK drug concentration in kidney (mg/L) 
CVK = AK/(VK*PK) 
d/dt(AUCCK) = CK  ; AUCCK AUC of drug concentration in kidney (mg*h/L) 
init AUCCK = 0 
 
; Penicillin urinary excretion 
Rurine = Kurine*CVK  
d/dt(Aurine) = Rurine 
init Aurine = 0 
 
; Penicillin in muscle compartment, flow-limited model 
RM = QM*(CAfree-CVM)  ; RM the changing rate of the amount of drug in muscle (mg/h)  
d/dt(AM) = RM   ; AM amount of the drug in muscle (mg) 
init AM = 0 
CM = AM/VM   ; CM drug concentration in muscle (mg/L) 
CVM = AM/(VM*PM) 
d/dt(AUCCM) = CM 
init AUCCM = 0 
 
; Penicillin in fat compartment, flow-limited model 
RF = QF*(CAfree-CVF)   ; RF the changing rate of the amount of drug in fat (mg/h) 
d/dt(AF) = RF   ; AF amount of the drug in fat (mg) 
init AF = 0 
CF = AF/VF   ; CF drug concentration in fat (mg/L) 
CVF = AF/(VF*PF) 
d/dt(AUCCF) = CF  ; AUCCF AUC of drug concentration in fat (mg*h/L) 
init AUCCF = 0 
 
; Penicillin in the compartment of rest of body, flow-limited model 
RR = QR*(CAfree-CVR)  ; Rrest the changing rate of the amount of drug in the rest of the body (mg/h) 
d/dt(AR) = RR   ; Arest amount of the drug in the rest of the body (mg) 
init AR = 0 
CR = AR/VR   ; Crest drug concentration in the rest of the body (mg/L) 
CVR = AR/(VR*Prest) 
d/dt(AUCCR) = CR  ; AUCCrest AUC of drug concentration in the rest of the body (mg*h/L) 
init AUCCR = 0 
 
; Penicillin in lung compartment, flow-limited model 
RLu = QLu*(CV-CVLu)  ; RLu the changing rate of the amount of drug in the lung (mg/h) 
d/dt(ALu) = RLu   ; ALu amount of the drug in the lung (mg) 
init ALu = 0 
CLu = ALu/VLu   ; CLu drug concentration in the rest of the lung (mg/L) 
CVLu = ALu/(VLu*PLu) 
d/dt(AUCCLu) = CLu  ; AUCCLu AUC of drug concentration in the lung (mg*h/L) 
init AUCCLu = 0 

{Mass balance equations} 
Qbal = QC-QM-QR-QF-QK-QL 
Tmass = ABlood+AM+ALu+AR+AF+AK+AL+Aurine+Amet 
Input = Absorbim+Absorbsc 
Bal = Input-Tmass 

Population PBPK Model Code 
Note: The Berkeley Madonna model code below is a population physiologically based pharmacokinetic 
(PBPK) model for procaine penicillin G in cattle and swine. Parameter values used in the model code are 
for procaine penicillin G in cattle. All parameter values used for population model in swine and cattle are 
summarized in Tables 2-3. Physiological parameter values of cattle and swine reported in our earlier 
paper by Lin et al. (2016) were for an average animal collected from previously published PBPK models. 
In the present penicillin G modeling work, in order to conduct population analysis, we need to have 
distributions of all parameters, so we conducted more extensive literature search on the physiological 
parameters of cattle and swine. As a result, some of the physiological parameters have been updated in 
this study. The updated physiological parameter values will be used in our subsequent models. Overall, 
the mean value of each physiological parameter is still quite close to the value reported in Lin et al. 
(2016) for both cattle and swine. 

{  
Monte Carlo Analysis based on Penicillin PBPK model for Cattle (flow-limited model, linear metabolism equation, 
plasma protein binding) 
The PBPK model code is based on the Oxytetracycline.mmd from Zhoumeng Lin 
} 
 
METHOD RK4 
 
STARTTIME = 0 
STOPTIME= 250  ; h, 24 
DT = 0.000125 
DTOUT = 0.1 
 
{Physiological Parameters} 
; Blood Flow Rates 
QCC = 5.970    ; L/h/kg, Cardiac Output (1960 Doyle) 
 
; Fracion of blood flow to organs (unitless)  
QLC =  0.405   ; Fraction of blood flow to the liver (1996 Lescoat, 1960 Dlyle) 
QKC = 0.090   ; Fraction of blood flow to the kidneys (2016 Lin) 
QMC = 0.180   ; Fraction of blood flow to the muscle (2016 Lin) 
QFC = 0.080   ; Fraction of blood flow to the fat (2016 Lin) 
QLuC = 1   ; Fraction of blood flow to the lung considered to be 1 
QrestC = 0.245   ; Fraction of blood flow to the rest of body (total sum equals to 1) 
 
; Tissue Volumes  
BW = 299.96   ; Body Weight (kg) (2011 Mirzaei) 
 
; Fractional organ tissue volumes (unitless) 
VLC = 0.014   ; Fractional liver tissue (1933 Swett) 
VKC = 0.002   ; Fractional kidney tissue (1933 Swett) 
VFC = 0.150   ; Fractional fat tissue (2016 Lin, 2014 Leavens) 
VMC = 0.270   ; Fractional muscle tissue (2016 Lin, 2014 Leavens) 
VLuC = 0.008   ; Fractional lung tissue (2016 Lin, 2014 Leavens) 
VvenC = 0.030   ; Venous blood volume, fraction of blood volume (2016 Lin; 2008 Leavens) 
VartC = 0.010   ; Arterial blood volume, fraction of blood volume (2016 Lin; 2008 Leavens) 
VrestC = 0.516   ; Fractional rest of body (total sum equals to 1) 
 
{Mass Transfer Parameters (Chemical-Specific Parameters)} 
; Partition Coefficients (PC, tissue:plasma) 
PL = 3.000   ; Liver:plasma PC (0.157, Tsuji et al., 1983, Table 4, in rats) 

PK = 2.500   ; Kidney:plasma PC (3.70, Tsuji et al., 1983, Table 4, in rats) 
PM = 0.300   ; Muscle:plasma PC (0.062, Tsuji et al., 1983, Table 4, in rats) 
PF = 0.040   ; Fat:plasma PC (0.062, adapted from muscle partition coefficient)  
PLu = 0.180   ; Lung:plasma PC (0.157, Tsuji et al., 1983, Table 4, in rats) 
Prest = 0.479   ; Rest of body:plasma PC (Cao et al. 2012, Table 1, estimated value in human) 
 
{Kinetic Constants} 
; IM Absorption Rate Constants 
Kim = 0.070   ; /h, IM absorption rate constant  
Frac = 0.600 
Kdiss = 1e-5    ; /h 
 
; SC Absorption Rate Constants 
Ksc = 0.020    ; /h, SC absorption rate constant 
Fracsc = 0.700 
Kdisssc = 1e-4 
 
; Percentage Plasma Protein Binding unitless 
PB = 0.483   ; Percentage of drug bound to plasma proteins (1965 Keen) 
Free = 1-PBm     ; Percentage of drug not bound to plasma protein 
 
{Metabolic Rate Constant} 
KmC = 0.0025   ; /h/kg, metabolic rate constant 
 
; Urinary Elimination Rate Constants 
KurineC = 0.450   ; L/h/kg 
 
{Parameters for Various Exposure Scenarios} 
PDOSEsc = 0   ; (mg/kg) 
PDOSEim = 6.5   ; (mg/kg) 
 
{Variances of Parameters} 
QCC_sd = 1.99   ; Standard deviation of QCC 
QLC_sd = 0.1942   ; Standard deviation of QLC 
QKC_sd = 0.027    ; Standard deviation of QKC 
QMC_sd = 0.054   ; Standard deviation of QMC 
QFC_sd = 0.024    ; Standard deviation of QFC 
QrestC_sd = 0.0736  ; Standard deviation of QrestC 
BW_sd = 46.18    ; Standard deviation of Body Weight 
VLC_sd = 1.63e-3  ; Standard deviation of VLC 
VKC_sd = 4.321e-4   ; Standard deviation of VKC 
VMC_sd = 8.1e-2   ; Standard deviation of VMC 
VFC_sd = 4.5e-2    ; Standard deviation of VFC 
VLuC_sd = 1.696e-3  ; Standard deviation of VLuC 
VrestC_sd = 0.1548   ; Standard deviation of VrestC 
VvenC_sd = 8.88e-3   ; Standard deviation of VvenC 
VartC_sd = 3.12e-3   ; Standard deviation of VartC 
PL_sd = 0.6     ; Standard deviation of PL 
PK_sd = 0.50    ; Standard deviation of PK 
PM_sd =0.060   ; Standard deviation of PM 
PF_sd = 8e-3     ; Standard deviation of PF 
PLu_sd = 3.6e-2     ; Standard deviation of PLu 
Prest_sd = 9.58e-2   ; Standard deviation of Prest 
Kim_sd = 2.10e-2   ; Standard deviation of Kim 
Frac_sd = 0.06   ; Standard deviation of Frac 
Kdiss_sd = 3e-6    ; Standard deviation of Kdiss 

Ksc_sd = 6e-3    ; Standard deviation of Ksc 
Fracsc_sd = 0.07   ; Standard deviation of Fracsc 
Kdisssc_sd = 3e-5   ; Standard deviation of Kdisssc 
PB_sd = 0.1449    ; Standard deviation of PB 
KmC_sd = 7.5e-4    ; Standard deviation of KmC 
KurineC_sd = 0.135    ; Standard deviation of KurineC 
 
{Generation of Parameters based on Normal Distribution} 
; Generation of Parameters based on Normal Distribution 
init QCCm = Normal(QCC, QCC_sd)   ; Generation of the QCCm based on normal distribution 
init QLCm = Normal(QLC, QLC_sd)    ; Generation of the QLCm based on normal distribution 
init QKCm = Normal(QKC, QKC_sd)    ; Generation of the QKCm based on normal distribution  
init QFCm = Normal(QFC, QFC_sd)    ; Generation of the QFCm based on normal distribution  
init QMCm = Normal(QMC, QMC_sd)    ; Generation of the QMCm based on normal distribution  
init QrestCm = Normal(QrestC, QrestC_sd)   ; Generation of the QrestCm based on normal distribution 
init BWm = Normal(BW, BW_sd)   ; Generation of the BWm based on normal distribution 
init VLCm = Normal(VLC, VLC_sd)    ; Generation of the VLCm based on normal distribution 
init VKCm = Normal(VKC, VKC_sd)    ; Generation of the VKCm based on normal distribution 
init VMCm = Normal(VMC, VMC_sd)   ; Generation of the VMCm based on normal distribution 
init VFCm = Normal(VFC, VFC_sd)     ; Generation of the VFCm based on normal distribution 
init VLuCm = Normal(VLuC, VLuC_sd)    ; Generation of the VLuCm based on normal distribution 
init VrestCm = Normal(VrestC, VrestC_sd)  ; Generation of the VrestCm based on normal distribution 
init VvenCm = Normal(VvenC, VvenC_sd)   ; Generation of the VvenCm based on normal distribution 
init VartCm = Normal(VartC, VartC_sd)    ; Generation of the VvartCm based on normal distribution 
 
; Assignment of the Values to Parameters 
next QCCm = QCCm    ; Assignment of the first created value to QCCm, without this 
step QCCm will change at each integration time step 
next BWm=BWm ; 
 
; Creation of Adjust Factor 
AdjustF = QLCm+QKCm+QFCm+QMCm+QrestCm     ; Adjust factor to keep the 
sum of blood flow fractions to 1 
AdjustF1=VLCm+VKCm+VMCm+VFCm+VLuCm+VrestCm+VvenCm+VartCm ; Adjustment factor to 
make sure the sum of fractions of organ tissue volumes to be 1 
 
; Creation of Adjusted Parameters 
next QLCm = QLCm/AdjustF    ; Adjustment of QLCm based on the adjust factor  
next QKCm = QKCm/AdjustF    ; Adjustment of QKCm 
next QFCm = QFCm/AdjustF    ; Adjustment of QFCm 
next QMCm = QMCm/AdjustF    ; Adjustment of QMCm 
next QrestCm = QrestCm/AdjustF    ; Adjustment of QrestCm 
next VLCm=VLCm/AdjustF1     ; Adjustment of VLCm based on the adjust factor  
next VKCm=VKCm/AdjustF1     ; Adjustment of VKCm based on the adjust factor  
next VMCm=VMCm/AdjustF1     ; Adjustment of VMCm based on the adjust factor  
next VFCm=VFCm/AdjustF1     ; Adjustment of VFCm based on the adjust factor  
next VLuCm=VLuCm/AdjustF1     ; Adjustment of VLuCm based on the adjust factor  
next VrestCm=VrestCm/AdjustF1     ; Adjustment of VrestCm based on the adjust factor  
next VvenCm=VvenCm/AdjustF1     ; Adjustment of VvenCm based on the adjust factor  
next VartCm=VartCm/AdjustF1     ; Adjustment of VartCm based on the adjust factor  
 
{Lognormal Transformation of Parameters} 
PL_ln = logn(PL^2/(PL_sd^2+PL^2)^0.5)    ; Lognormal transformation of PL values 
PL_lnsd = (logn(1+PL_sd^2/PL^2))^0.5 
PK_ln = logn(PK^2/(PK_sd^2+PK^2)^0.5)    ; Lognormal transformation of PK values 
PK_lnsd = (logn(1+PK_sd^2/PK^2))^0.5 

PM_ln = logn(PM^2/(PM_sd^2+PM^2)^0.5)     ; Lognormal transformation of PM values 
PM_lnsd = (logn(1+PM_sd^2/PM^2))^0.5 
PF_ln = logn(PF^2/(PF_sd^2+PF^2)^0.5)     ; Lognormal transformation of PF values 
PF_lnsd = (logn(1+PF_sd^2/PF^2))^0.5 
PLu_ln = logn(PLu^2/(PLu_sd^2+PLu^2)^0.5)     ; Lognormal transformation of PLu values 
PLu_lnsd = (logn(1+PLu_sd^2/PLu^2))^0.5 
Prest_ln = logn(Prest^2/(Prest_sd^2+Prest^2)^0.5)    ; Lognormal transformation of Prest values 
Prest_lnsd = (logn(1+Prest_sd^2/Prest^2))^0.5 
Kim_ln = logn(Kim^2/(Kim_sd^2+Kim^2)^0.5)   ; Lognormal transformation of Kim value 
Kim_lnsd = (logn(1+Kim_sd^2/Kim^2))^0.5 
Frac_ln = logn(Frac^2/(Frac_sd^2+Frac^2)^0.5)     ; Lognormal transformation of Frac value 
Frac_lnsd = (logn(1+Frac_sd^2/Frac^2))^0.5 
Kdiss_ln = logn(Kdiss^2/(Kdiss_sd^2+Kdiss^2)^0.5)   ; Lognormal transformation of Kdiss value 
Kdiss_lnsd = (logn(1+Kdiss_sd^2/Kdiss^2))^0.5 
Ksc_ln = logn(Ksc^2/(Ksc_sd^2+Ksc^2)^0.5)    ; Lognormal transformation of Ksc value 
Ksc_lnsd = (logn(1+Ksc_sd^2/Ksc^2))^0.5 
Fracsc_ln = logn(Fracsc^2/(Fracsc_sd^2+Fracsc^2)^0.5)    ; Lognormal transformation of Fracsc value 
Fracsc_lnsd = (logn(1+Fracsc_sd^2/Fracsc^2))^0.5 
Kdisssc_ln = logn(Kdisssc^2/(Kdisssc_sd^2+Kdisssc^2)^0.5)  ; Lognormal transformation of Kdisssc value 
Kdisssc_lnsd = (logn(1+Kdisssc_sd^2/Kdisssc^2))^0.5 
PB_ln = logn(PB^2/(PB_sd^2+PB^2)^0.5)      ; Lognormal transformation of PB 
PB_lnsd = (logn(1+PB_sd^2/PB^2))^0.5 
KmC_ln = logn(KmC^2/(KmC_sd^2+KmC^2)^0.5)    ; Lognormal transformation of KmC 
KmC_lnsd = (logn(1+KmC_sd^2/KmC^2))^0.5 
KurineC_ln = logn(KurineC^2/(KurineC_sd^2+KurineC^2)^0.5) ; Lognormal transformation of KurineC 
KurineC_lnsd = (logn(1+KurineC_sd^2/KurineC^2))^0.5   
 
{Creation of Parameters based on Lognormal Distribution} 
init PLm = exp(Normal(PL_ln, PL_lnsd)) next PLm = PLm       ; Generation of 
PLm based on lognormal distribution 
init PMm = exp(Normal(PM_ln, PM_lnsd)) next PMm = PMm      ; Generation of 
PMm  
init PFm = exp(Normal(PF_ln, PF_lnsd)) next PFm = PFm      ; Generation of PFm  
init PKm = exp(Normal(PK_ln, PK_lnsd)) next PKm = PKm      ; Generation of PKm 
init PLum = exp(Normal(PLu_ln, PLu_lnsd)) next PLum = PLum      ; Generation of 
PLum 
init Prestm = exp(Normal(Prest_ln, Prest_lnsd)) next Prestm = Prestm     ; Generation of Prestm 
init Kimm = exp(Normal(Kim_ln, Kim_lnsd)) next Kimm = Kimm      ; Generation of 
Kimm 
init Fracm = exp(Normal(Frac_ln, Frac_lnsd)) next Fracm = Fracm     ; Generation of Fracm 
init Kdissm = exp(Normal(Kdiss_ln, Kdiss_lnsd)) next Kdissm = Kdissm     ; Generation of 
Kdissm 
init Kscm = exp(Normal(Ksc_ln, Ksc_lnsd)) next Kscm = Kscm      ; Generation of 
Kscm 
init Fracscm = exp(Normal(Fracsc_ln, Fracsc_lnsd)) next Fracscm = Fracscm   ; Generation of Fracscm 
init Kdissscm = exp(Normal(Kdisssc_ln, Kdisssc_lnsd)) next Kdissscm = Kdissscm   ; Generation of Kdissscm 
init PBm = exp(Normal(PB_ln, PB_lnsd)) next PBm = PBm       ; Generation of PBm 
init KmCm = exp(Normal(KmC_ln, KmC_lnsd)) next KmCm = KmCm     ; Generation of 
KmCm 
init KurineCm = exp(Normal(KurineC_ln, KurineC_lnsd)) next KurineCm = KurineCm   ; Generation of KurineCm 
 
{limit the parameter values within the lower and upper bounds} 
limit BWm >= 209.45 
limit BWm <= 390.464 
limit QCCm >= 2.07 
limit QCCm <= 9.87 

limit VartCm >= 0.004 
limit VartCm <= 0.017 
limit VvenCm >= 0.012 
limit VvenCm <= 0.047 
limit VLCm >= 0.010 
limit VLCm <= 0.017 
limit VKCm >= 0.002 
limit VKCm <= 0.003 
limit VMCm >= 0.111 
limit VMCm <= 0.429 
limit VFCm >= 0.062 
limit VFCm <= 0.238 
limit VLuCm >= 0.005 
limit VLuCm <= 0.011 
limit VrestCm >=0.213 
limit VrestCm <=0.819 
limit QLCm >=0.024 
limit QLCm <=0.785 
limit QKCm >=0.037 
limit QKCm <=0.143 
limit QMCm >=0.074 
limit QMCm <=0.286 
limit QFCm >=0.033 
limit QFCm <=0.127 
limit QrestCm >=0.101 
limit QrestCm <=0.390 
limit Kimm >=0.038 
limit Kimm <= 0.119 
limit Fracm >= 0.491 
limit Fracm <= 0.726 
limit Kdissm >=5.388e-6 
limit Kdissm <= 1.703e-5 
limit Kscm >= 0.011 
limit Kscm <= 0.034 
limit Fracscm >= 0.573 
limit Fracscm <= 0.847 
limit Kdissscm >= 5.388e-5 
limit Kdissscm <= 1.703e-4 
limit PLm >= 1.995 
limit PLm <= 4.337 
limit PKm >= 1.663 
limit PKm <= 3.614 
limit PMm >= 0.2 
limit PMm <= 0.434 
limit PFm >= 0.027 
limit PFm <= 0.058 
limit PLum >= 0.12 
limit PLum <= 0.26 
limit Prestm >= 0.319 
limit Prestm <= 0.692 
limit KmCm >=0.001 
limit KmCm <=0.004 
limit PBm >=0.260 
limit PBm <=0.822 
limit KurineCm >= 0.242 
limit KurineCm <= 0.766 

{Cardiac output and blood flow to tissues (L/h)} 
QC = QCCm*BWm    ; Cardiac output 
QL = QLCm*QC    ; Liver 
QK = QKCm*QC    ; Kidney 
QF = QFCm*QC    ; Fat 
QM = QMCm*QC    ; Muscle 
QLu = QLuC*QC    ; Lung 
QR = QrestCm*QC    ; Rest of body 
 
{Tissue volumes (L)} 
VL = VLCm*BWm    ; Liver 
VK = VKCm*BWm    ; Kidney 
VF = VFCm*BWm    ; Fat 
VM = VMCm*BWm    ; Muscle 
VLu = VLuCm*BWm    ; Lung 
VR = VrestCm*BWm    ; Rest of body 
Vven = VvenCm*BWm    ; Venous Blood 
Vart = VartCm*BWm    ; Arterial Blood 
 
; Metabolism Rate Constants 
Kmet = KmCm*BWm 
 
; Urinary Elimination Rate Constants 
Kurine = KurineCm*BWm 
 
{Dosing} 
; Dosing caculation based on BW 
DOSEsc = PDOSEsc*BWm  ; (mg) 
DOSEim = PDOSEim*BWm  ; (mg) 
 
; Dosing, repeated doses 
tinterval = 24   ; Varied dependent on the exposure paradigm (h) 
Tdoses = 5    ; times for multiple oral gavage 
 
dosingperiod = if time < Tdoses*tinterval-DT then 1 else 0 
 
; Dosing, IM, intramuscular 
Rinputim = pulse(DOSEim,0,tinterval)*dosingperiod 
Rpenim = Rinputim*(1-Fracm); 
Rppgim =  Rinputim*Fracm; 
Rim = Kimm*Amtsiteim 
d/dt(Absorbim) = Rim 
init Absorbim = 0 
d/dt(Amtsiteim) = Rpenim- Rim + Kdissm* DOSEppgim 
init Amtsiteim = 0 
d/dt(DOSEppgim) = Rppgim-Kdissm* DOSEppgim 
init DOSEppgim = 0 
 
; Dosing, SC, subcutaneous 
Rinputsc = pulse(DOSEsc,0,tinterval)*dosingperiod 
Rpensc = Rinputsc*(1-Fracscm); 
Rppgsc =  Rinputsc*Fracscm; 
Rsc = Kscm*Amtsitesc 
d/dt(Absorbsc) = Rsc 
init Absorbsc = 0 

d/dt(Amtsitesc) = Rpensc- Rsc + Kdissscm* DOSEppgsc 
init Amtsitesc = 0 
d/dt(DOSEppgsc) = Rppgsc-Kdissscm* DOSEppgsc 
init DOSEppgsc = 0 
 
{Penicillin distribution in each compartment} 
; Penicillin in venous blood compartment 
RV = (QL*CVL+QK*CVK+QF*CVF+QM*CVM+QR*CVR+Rsc+Rim)-QC*CV; RV the changing rate in the 
venous blood (mg/h) 
d/dt(AV) = RV   ; AV the amount of the drug in the venous blood (mg) 
init AV = 0 
CV = AV/Vven   ; CV drug concentration in the venous blood (mg/L) 
RA = QC*(CVLu-CAfree)  ; RA the changing rate in the arterial blood (mg/h) 
d/dt(AA) = RA 
init AA = 0   ; AA the amount of the drug in the arterial blood (mg) 
CA = AA/Vart   ; CAfree concentration of unbound drug in the arterial blood (mg/L) 
CAfree = CA*Free 
d/dt(AUCCV) = CV  ; AUCCV AUC of drug concentration in the venous blood (mg*h/L) 
init AUCCV = 0 
 
ABlood = AA+AV 
 
; Penicillin in liver compartment, flow-limited model 
RL = QL*(CAfree-CVL)-Rmet  ; RL the changing rate of the amount of drug in liver (mg/h) 
d/dt(AL) = RL    ; AL amount of drug in liver (mg)  
init AL = 0   
CL = AL/VL   ; CL drug concentration in liver (mg/L) 
CVL= AL/(VL*PLm)  ; CVL drug concentration in venous blood from liver (mg/L) 
d/dt(AUCCL) = CL  ; AUCCL area under the curve of drug concentration in liver (mg*h/L) 
init AUCCL = 0 
 
; Metabolism of Penicillin in liver compartment 
Rmet = Kmet*CL*VL  ; Rmet the metabolic rate in liver (mg/h) 
d/dt(Amet) = Rmet  ; Amet the amount of drug metabolized in liver (mg) 
init Amet = 0 
 
; Penicillin in kidney compartment, flow-limited model 
RK = QK*(CAfree-CVK)-Rurine ; RK the changing rate of the amount of drug in kidney (mg/h) 
d/dt(AK) = RK   ; AK amount of drug in kidney (mg) 
init AK = 0   
CK = AK/VK   ; CK drug concentration in kidney (mg/L) 
CVK = AK/(VK*PKm) 
d/dt(AUCCK) = CK  ; AUCCK AUC of drug concentration in kidney (mg*h/L) 
init AUCCK = 0 
 
; Penicillin urinary excretion 
Rurine = Kurine*CVK 
d/dt(Aurine) = Rurine 
init Aurine = 0 
 
; Penicillin in muscle compartment, flow-limited model 
RM = QM*(CAfree-CVM)  ; RM the changing rate of the amount of drug in muscle (mg/h)  
d/dt(AM) = RM   ; AM amount of the drug in muscle (mg) 
init AM = 0 
CM = AM/VM   ; CM drug concentration in muscle (mg/L) 
CVM = AM/(VM*PMm) 

d/dt(AUCCM) = CM 
init AUCCM = 0 
 
; Penicillin in fat compartment, flow-limited model 
RF = QF*(CAfree-CVF)   ; RF the changing rate of the amount of drug in fat (mg/h) 
d/dt(AF) = RF   ; AF amount of the drug in fat (mg) 
init AF = 0 
CF = AF/VF   ; CF drug concentration in fat (mg/L) 
CVF = AF/(VF*PFm) 
d/dt(AUCCF) = CF  ; AUCCF AUC of drug concentration in fat (mg*h/L) 
init AUCCF = 0 
 
; Penicillin in the compartment of rest of body, flow-limited model 
RR = QR*(CAfree-CVR)   ; Rrest the changing rate of the amount of drug in the rest of the body (mg/h) 
d/dt(AR) = RR   ; Arest amount of the drug in the rest of the body (mg) 
init AR = 0 
CR = AR/VR   ; Crest drug concentration in the rest of the body (mg/L) 
CVR = AR/(VR*Prestm) 
d/dt(AUCCR) = CR  ; AUCCrest AUC of drug concentration in the rest of the body (mg*h/L) 
init AUCCR = 0 
 
; Penicillin in lung compartment, flow-limited model 
RLu = QLu*(CV-CVLu)  ; RLu the changing rate of the amount of drug in the lung (mg/h) 
d/dt(ALu) = RLu   ; ALu amount of the drug in the lung (mg) 
init ALu = 0 
CLu = ALu/VLu   ; CLu drug concentration in the rest of the lung (mg/L) 
CVLu = ALu/(VLu*PLum) 
d/dt(AUCCLu) = CLu  ; AUCCLu AUC of drug concentration in the lung (mg*h/L) 
init AUCCLu = 0 
 
{Mass balance equations} 
Qbal = QC-QM-QR-QF-QK-QL 
Tmass = ABlood+AM+ALu+AR+AF+AK+AL+Aurine+Amet 
Input = Absorbim+Absorbsc 
Bal = Input-Tmass  

